

RFC 2234

Sözdizimi Belirtileri için Arttırımlı BNF: ABNF

Düzenleyen:
D. Crocker
Internet Mail Consortium

Düzenleyen:
P. Overell
Demon Internet Ltd.

Çeviren:
Nilgün Belma Bugüner
belgeler.org

Ekim 2006

Özet

Arpanet'in ilk günlerinde, her belirtim kendi ABNF belirtimini içeriyordu. Buna eposta belirtileri de dahildi. RFC822, RFC733'ten yaptığı ABNF alıntıları ile geldi. Bu belge belli bir başvuru kaynağı olmak üzere bu tanımları ayırmaktadır. Baştan belirtelim, ayrıca bazı değişiklikler ve genişletmeler de içerir.

Bu çeviri, RFC Türkçe çevirilerinde kullanılan ABNF temel belirtimini içerir.

Konu Başlıkları

1. Giriş	3
2. Kural Tanımı	3
2.1. Kural Adlandırma	3
2.2. Kural Biçimi	3
2.3. Uç Değerler	4
2.4. Harici Kodlamalar	5
3. İşleçler	5
3.1. Ulama — kural1 kural2	5
3.2. İkame — kural1 / kural2	5
3.3. İkame Eklemeleri — kural1=/ kural2	6
3.4. İkame Değer Aralıkları — %c##-##	6
3.5. Dizilim Grubu — (kural1 kural2)	6
3.6. Değişken Yineleme — *kural	7
3.7. Belirgin Yineleme — nkural	7
3.8. Seçimlik Dizilim — [kural]	7
3.9. ; Açıklama	7
3.10. İşleç Önceliği	7
4. ABNF'nin ABNF Tanımı	8
5. Güvenlikle İlgili Hususlar	9
A. NÜVE	10
B. Teşekkür	11
C. Kaynakça	11
D. Yazar Adresleri	11
E. Tam Telif Hakkı Beyanı	12

Geçmiş

1.0 İlk çeviri	Ekim 2006	NBB
Standart Aday Adayı Özgün sürüm	Kasım 1997	DC ve PO

Sürüm Bilgileri

Ağ Çalışma Grubu
Açıklama İsteği: 2234
Yerini alan RFC: 4234
Durumu: Standart Aday Adayı

Yasal Uyarı

RFC'lerin yazarlarının hakları [BCP 78^{\(B1\)}](#) ile düzenlenmiştir. Dolayısıyla RFC çevirilerinin çevirmenlerinin haklarını da BCP 78'in düzenlediği kabul edilmiştir.

Bu belge [IETF^{\(B3\)}](#) tarafından yayınlanan resmi RFC 2234'ün **gayriresmi** çevirisidir ve aslının yerine kullanılamaz. Bu çevirinin hiçbir bağlamda ya da koşulda hükmü yoktur. Bu çeviri, anadili Türkçe olan internet kullanıcılarının bu RFC hakkında fikir edinebilmelerini sağlamak amacıyla hazırlanmıştır.

BU BELGE "ÜCRETSİZ" OLARAK RUHSATLANDIĞI İÇİN, İÇERDİĞİ BİLGİLER İÇİN İLGİLİ KANUNLARIN İZİN VERDİĞİ ÖLÇÜDE HERHANGİ BİR GARANTİ VERİLMEMEKTEDİR. AKSİ YAZILI OLARAK BELİRTİLMEDİĞİ MÜDDETÇE TELİF HAKKI SAHİPLERİ VE/VEYA BAŞKA ŞAHISLAR BELGEYİ "OLDUĞU GİBİ", AŞIKAR VEYA ZIMNEN, SATILABİLİRLİĞİ VEYA HERHANGİ BİR AMACA UYGUNLUĞU DA DAHİL OLMAK ÜZERE HİÇBİR GARANTİ VERMEKSİZİN DAĞITMAKTADIRLAR. BİLGİNİN KALİTESİ İLE İLGİLİ TÜM SORUNLAR SİZE AİTTİR. HERHANGİ BİR HATALI BİLGİDEN DOLAYI DOĞABİLECEK OLAN BÜTÜN SERVİS, TAMİR VEYA DÜZELTME MASRAFLARI SİZE AİTTİR.

İLGİLİ KANUNUN İCBAR ETTİĞİ DURUMLAR VEYA YAZILI ANLAŞMA HARİCİNDE HERHANGİ BİR ŞEKİLDE TELİF HAKKI SAHİBİ VEYA YUKARIDA İZİN VERİLDİĞİ ŞEKİLDE BELGEYİ DEĞİŞTİREN VEYA YENİDEN DAĞITAN HERHANGİ BİR KİŞİ, BİLGİNİN KULLANIMI VEYA KULLANILAMAMASI (VEYA VERİ KAYBI OLUŞMASI, VERİNİN YANLIŞ HALE GELMESİ, SİZİN VEYA ÜÇÜNCÜ ŞAHISLARIN ZARARA UĞRAMASI VEYA BİLGİLERİN BAŞKA BİLGİLERLE UYUMSUZ OLMASI) YÜZÜNDEN OLUŞAN GENEL, ÖZEL, DOĞRUDAN YA DA DOLAYLI HERHANGİ BİR ZARARDAN, BÖYLE BİR TAZMİNAT TALEBİ TELİF HAKKI SAHİBİ VEYA İLGİLİ KİŞİYE BİLDİRİLMİŞ OLSA DAHI, SORUMLU DEĞİLDİR.

Tüm telif hakları aksi özellikle belirtilmediği sürece sahibine aittir. Belge içinde geçen herhangi bir terim, bir ticari isim ya da kuruma itibar kazandırma olarak algılanmamalıdır. Bir ürün ya da markanın kullanılmış olması ona onay verildiği anlamında görülmemelidir.

Bu Belgenin Durumu

Bu belge Genel Ağ standardı olmak üzere standartlaşma aşamasına girmiş bir protokolü belirtir ve geliştirilmesi için Genel Ağ topluluğundan üzerinde tartışılmasını ve önerilerde bulunulmasını talep eder. Bu protokolün durumu ve standartlaşma aşamalarındaki seviyesi için lütfen "Genel Ağ Resmi Protokol Standartları" (STD 1) belgesinin şu an ki sürümüne bakınız. Bu belgenin dağıtımına sınırlama getirilmemiştir.

Copyright © The Internet Society (1997). All Rights Reserved.

1. Giriş

Genel Ağ teknik belirtimlerinde yazarlarının yararlı bulduğu her gösterim kullanılabilir ve bu yüzden sıklıkla bir biçim sözdizimi tanımlama gereği duyulur. Yıllardan sonra, Arttırımlı BNF (ABNF) adı verilen, Backus–Naur Formunun değiştirilmiş bir sürümü çoğu Genel Ağ belirtiminde yaygın olarak kullanılır olmaya başladı. Makul gösterimsel gücü sayesinde bütünleşiklik ile basitliği dengeler. Arpanet'in ilk günlerinde, her belirtim kendi ABNF belirtimini içerirdi. Buna eposta belirtimleri de dahildi. RFC822, RFC733'ten yaptığı ABNF alıntıları ile geldi. Bu belge belli bir başvuru kaynağı olmak üzere bu tanımları ayırmaktadır. Baştan belirtelim, ayrıca bazı değişiklikler ve genişletmeler de içerir.

Standart BNF ile ABNF arasındaki farklar olarak, isimleme kuralları, yinelemeler, diğer şekiller, bağımsız sıralama ve değer aralıkları gibi gösterimler sayılabilir. [NÜVE](#) (sayfa: 10) bölümü çeşitli Genel Ağ belirtimleri arasında ortak türden bir çekirdek sözel çözümleyici için kural tanımları ve kodlama içerir. Bir uzlaşım olarak sağlanmış ve diğer hususlarda bu belgede tanımlanan ötedilden ve biçimsel durumundan ayrı tutulmuştur.

Bu çeviri, RFC Türkçe çevirilerinde kullanılan ABNF temel belirtimini içerir.

2. Kural Tanımı

2.1. Kural Adlandırma

Bir kural ismi basitçe ismin kendisidir; yani, bir alfabetik karakterle başlayan harfler, rakamlar ve tire imlerinden oluşan bir karakter dizisidir.



Bilgi

Kural isimleri harf büyüklüğüne duyarlı değildir.

<kuraladı>, <Kuraladı>, <KURALADI>, <kURaLaDı>, hepsi aynı kuralı ifade eder.

Özgün BNF'nin aksine, açılı ayraçlar ("<", ">") gerekli değildir. Bununla birlikte, açılı ayraçlar, varlıkları bir kural ismi kullanımının ayırtedilmesini kolaylaştıracağı her yerde bir kural ismini sarmalamakta kullanılabilir. Bu kullanım, genellikle, serbest düzyazıda kural isimlerine atıf yapmakla veya kural isimleri boşluklarla ayrılmamış bir dizge olarak verildiğinde içinde yeralan kuralların farkedilmesini sağlamakla sınırlıdır; bir örneği aşağıda, yinelemenin açıklamasında görülebilir.

2.2. Kural Biçimi

Bir kural aşağıdaki sırayla tanımlanır:

```
isim = elemanlar crlf
```

Burada *<isim>* kuralın ismi, *<elemanlar>* başka kural veya kuralların isimleri ya da uç belirtimleri, *<crlf>* ise satırbaşı karakterini izleyen satırsonu karakterinden oluşan satırsonu belirtecidir. Eşit işareti kural ismini

tanımından ayırır. Elemanlar, bu belgede tanımlanmış çeşitli işleçlerle birleştirilmiş bir veya daha fazla kural ismi ve/veya değer tanımından oluşur.

Görsel rahatlık amacıyla kural tanımları sola yanaşıktır. Bir kural birden fazla satır gerektiriyorsa, devam satırları girintilenir. Sola hizalama ve girintileme ABNF kuralının ilk satırına görelidir ve belgenin sol kenar payıyla uyumlu olması gerekmez.

2.3. Uç Değerler

Kurallar bir uç değerler dizgesi olarak çözümlenir ve bunlara bazan karakterler adı verilir. ABNF'de bir karakter basitçe sıfır veya pozitif bir tamsayıdır. Belirli bağlamlarda, değerlerin özgün bir eşlemi (kodlaması) bir karakter kümesi olarak (ASCII gibi) belirtilir.

Uç değerler, bu karakterlerin taban yorumlarını açıkça belirten bir veya daha fazla sayısal karakterle belirtilir. Şimdilik şu tabanlar tanımlanmıştır:

```
b          =  ikilik
d          =  onluk
x          =  onaltılık
```

Dolayısıyla:

```
CR          =  %d13
ve
CR          =  %x0D
```

sırasıyla satırbaşı karakterinin onluk ve onaltılık değerlerini [*US-ASCII*] belirtir.

Böyle değerlerin birleşiminden oluşan değerler bir bütün olarak, değeri oluşturan karakterlerin arasına bir nokta (".") konarak belirtilir. Örnek:

```
CRLF        =  %d13.10
```

ABNF, dizgesel sabitlerin tırnaklar içine alınarak doğrudan belirtilebilmesine imkan tanır. Örnek:

```
komut       =  "komut dizgesi"
```

Dizgesel sabitler basılabilen karakterlerin bir birleşimi olarak yorumlanır.



Bilgi

ABNF dizgesel sabitleri harf büyüklüğüne duyarlı değildir ve bu dizgelerin karakter kümesi us-ascii'dir.

Örneğin,

```
kuralismi = "abc"
```

ve

```
kuralismi = "aBc"
```

"abc", "Abc", "aBc", "abC", "ABc", "aBC", "AbC" ve "ABC" ile eşleşir.

Harf büyüklüklerine duyarlı bir kuralın her karakteri tek tek belirtilir.

Örneğin,

```
kuralismi = %d97 %d98 %d99
```

veya

```
kuralismi = %d97.98.99
```

sadece küçük harfli dizgeyle ("abc") eşleşecektir.

2.4. Harici Kodlamalar

Uç değer karakterlerinin harici gösterimleri saklama ya da aktarım ortamının kısıtlamalarına tabidir. Bu nedenle, aynı ABNF tabanlı dilbilgisinin çok sayıda harici kodlaması olabilir, örneğin, biri 16 bitlik Unicode kullanılırken biri 7 bitlik US-ASCII ortamı için, diğeri ikil sekizli ortam için olabilir. *NÜVE* (sayfa: 10) bölümündeki tanımlar Genel Ağ'ın çoğunluğu için ortak olarak 7 bitlik US-ASCII ortamı için sağlanmışsa da kodlama ile ilgili ayrıntılar ABNF kapsamının dışındadır.

Harici kodlamayı sözdiziminden ayırmak suretiyle diğer kodlama ortamları aynı sözdizimi için kullanılabilecek şekilde tasarlanmıştır.

3. İşleçler

3.1. Ulama — *kural1 kural2*

Bir kural, bir dizi kural ismini yan yana getirerek aslında basit değerlerin bir dizgesini tanımlayabilir. Örneğin,

```
foo      = %x61      ; a
bar      = %x62      ; b
mumble   = foo bar foo
```

<mumble> kuralı küçük harfli dizge "aba" ile eşleşir.

Düzgün Doğrusal Boşluk

Ulama, ABNF çözümleme modelinin çekirdeğini oluşturur. Birbiri ardına gelen karakterlerden oluşan bir dizge ABNF ile tanımlanmış kurallara göre çözülür. Genel Ağ belirtileri için, düzgün doğrusal boşlukların (boşluk ve sekme karakterleri) doğrudan ya da dolaylı olarak ana oluşumların etrafına serpiştirilmesine izin vermekle ilgili bir geçmiş vardır (örneğin, özel karakterler veya atomik dizgeleri sınırlamak gibi amaçlar).



Bilgi

Bu ABNF belirtimi, düzgün doğrusal boşlukların dolaylı belirtimini içermez.

Ayraçlar veya dizge parçaları çevresinde düzgün doğrusal boşluklara izin verilmesini isteyen bir dilbilgisinin açıkça belirtmesi gerekir. Yüksek seviyeden kurallar arasında kullanmak üzere böyle boşluklarla ilgili "çekirdek" kurallar tanımlamak çoğunlukla kullanışlı olmaktadır. "Çekirdek" kurallar, bir sözel çözümleyici için veya ana kural kümesinin basit bir parçası olarak şekillenebilir.

3.2. İkameler — *kural1 / kural2*

Birbirleri yerine kullanılabilecek elemanlar bölü çizgisi ("/") ile ayrılır. Yani,

```
foo / bar
```

<foo> veya *<bar>*'dan biri kabul edilecektir anlamına gelir.



Bilgi

Alfabetik karakterler içeren bir tırnaklı dizge ikame karakterleri belirtmenin özel bir halidir ve içerdiği karakterlerle, küçük ve büyük harflerin bir karışımı olarak belirtildiği sırada, birleştirici dizgelerin bir kümesinin kurlsız gösterimi olarak yorumlanır.

3.3. İkame Eklemeleri — *kural1* = / *kural2*

Bazan bir ikame listesini dilimler halinde belirtmek uygun olur. Yani, birden fazla ikamesi olan bir kuralın ikameleri ek kural tanımları ile belirtilebilir. Aynı ana kural kümesinden türemiş bağımsız belirtiler (parametre listeleri şeklinde oluşan) olmadıkça bu kısmen kullanışlıdır. Bu eklemeli tanımlara ABNF'de şöyle izin verilir:

```
eskikural    =/ ek-ikameler
```

Şöyle bir kural kümesi,

```
kuralkümesi  = alt1 / alt2
```

```
kuralkümesi  =/ alt3
```

```
kuralkümesi  =/ alt4 / alt5
```

şu belirtilme aynıdır:

```
kuralkümesi  = alt1 / alt2 / alt3 / alt4 / alt5
```

3.4. İkame Değer Aralıkları — %c##-##

İkame sayısal değerler, değer aralığı için bir tire imi ("–") kullanılarak tek parça halinde belirtilebilir. Örneğin,

```
RAKAM        = %x30–39
```

şuna eşdeğerdir:

```
RAKAM        = "0" / "1" / "2" / "3" / "4" / "5" / "6" /  
               "7" / "8" / "9"
```

Ulanan sayısal değerlerle sayısal değer aralıkları aynı dizgede belirtilemez. Ulanma için bir sayısal değer noktalı gösterimle belirtilebilirken bir değer aralığını belirtmek için tire kullanılır. Bu nedenle, satır sonu dizilimleri arasında basılabilir bir karakterin belirtilmesi için belirtim şöyle olabilirdi:

```
krk-satırı   = %x0D.0A %x20–7E %x0D.0A
```

3.5. Dizilim Grubu — (*kural1* *kural2*)

Parantez içine alınmış elemanlar içerildikleri sıra değişmemek üzere tek bir eleman olarak ele alınırlar.

```
elem (foo / bar) blat
```

şeklinde bir tanım, (elem foo blat) veya (elem bar blat) ile eşleşecektir.

```
elem foo / bar blat
```

ise (elem foo) veya (bar blat) ile eşleşir.



Bilgi

İkamelerin çok sayıda kural ismi veya sabitinden oluştuğu durumlarda, düzgün okuma için "çıplak" ikamel-erden daha iyi olacaksa gruplama kullanımı daima tavsiye edilir.

Bu nedenle üstteki biçim yerine şu biçim önerilir:

```
(elem foo) / (bar blat)
```

Böylece arasına düzgün okuyanların tanımı yanlış yorumlaması önlenmiş olur.

Dizilim grubu gösterimi ayrıca serbest yazı içinde de kullanılır.

3.6. Değişken Yineleme — **kural*

"" işleci bir elemanın önüne konulduğunda yineleme belirtir. Örneğin,

```
<a>*<b>eleman
```

gösteriminde *<a>* ve ** isteğe bağlı ondalık değerler olmak üzere içinde en az *<a>* veya en çok ** sayıda eleman içerebilen oluşumlar tanımlanmaktadır.

Öntanımlı değerler sıfır ve sonsuz olmak üzere, **<eleman>* sıfır dahil her sayıya izin verirken, *1*<eleman>* en az 1, *3*3<eleman>* tam olarak 3 ve *1*2<eleman>* ise bir ya da ikisine izin verir.

3.7. Belirgin Yineleme — *nkural*

```
<n>eleman
```

kuralı

```
<n>*<n>eleman
```

kuralına eşdeğerdir. Yani, tam olarak *<N> <eleman>* oluşumu belirtir. Bu bakımdan, *2RAKAM* 2 rakamlı bir sayı belirtirken *3HARF* 3 harften oluşan bir dizgeyi belirtecektir.

3.8. Seçimlik Dizilim — *[kural]*

Köşeli ayraç içine alma, isteğe bağlı dizilim belirtir:

```
[foo bar]
```

ile

```
*1(foo bar).
```

eşdeğerdir.

3.9. ; Açıklama

Bir noktalı virgöl, satır sonuna kadar devam eden bir açıklamayı başlatır. Bu, belirtimle birlikte açıklamaları da içermenin basit bir yoludur.

3.10. İşleç Önceliği

Buraya kadar açıklanan mekanizmalar en yüksekten en düşüğe (en sıkıdan en gevşeye) aşağıdaki öncelik sıralamasına uyarlar:

```
Dizgeler, İsimler
Açıklamalar
Değer aralıkları
```

Yineleme
 Gruplama, İsteğe bağlılık
 Ulama
 İkame

İkame işlecinin kullanımı, ulamalarla karıştırıldığında karışıklığa yol açabilir.

Yine, açık ulama grupları yapmak için gruplama işlecinin kullanılması önerilir.

4. ABNF'nin ABNF Tanımı

Bu sözdizimi *NÜVE* (sayfa: 10) bölümündeki kuralları kullanır.

```

kurallistesi  = 1*( kural / (*d-boşluk s-st) )

kural         = kuralismi tanımı-gibi elemanlar s-st
                ; sonraki satır boşluklarla başlıyorsa
                ; Sonraki SaTırdada devam eder

kuralismi     = HARF *(HARF / RAKAM / "-")

tanımı-gibi   = *d-boşluk ("=" / "=/") *d-boşluk
                ; temel kural tanımı ve
                ; ikame eklemeleri

elemanlar     = ikame *d-boşluk

d-boşluk      = BOŞLUK / (s-st BOŞLUK)

s-st          = açıklama / CRLF
                ; açıklama veya satırsonu

açıklama      = ";" *(BOŞLUK / GÖRKRK) CRLF

ikame         = ulama
                (*d-boşluk "/" *d-boşluk ulama)

ulama         = yineleme *(1*d-boşluk yineleme)

yineleme      = [tekrarsayısı] eleman

tekrarsayısı  = 1*RAKAM / (*RAKAM "*" *RAKAM)

eleman        = kuralismi / grup / seçenek /
                krk-dğr / say-dğr / düz-dğr

grup          = "(" *d-boşluk ikame *d-boşluk ")"

seçenek       = "[" *d-boşluk ikame *d-boşluk "]"

```



```

krk-dğr      =  ÇTIRNAK *(%x20-21 / %x23-7E) ÇTIRNAK
                ; ÇTIRNAK içermeksizin ÇTIRNAK içine
                alınmış BOŞKRK ve GÖRKRK dizgeleri

say-dğr      =  "%" (ikil-dğr / on-dğr / onaltı-dğr)

ikil-dğr     =  "b" 1*BIT
                [ 1*("." 1*BIT) / ("-" 1*BIT) ]
                ; peşpeşe bit değerleri
                ; veya tek başına birler aralığı

on-dğr       =  "d" 1*RAKAM
                [ 1*("." 1*RAKAM) / ("-" 1*RAKAM) ]

onaltı-dğr   =  "x" 1*RAKAM16
                [ 1*("." 1*RAKAM16) / ("-" 1*RAKAM16) ]

düz-dğr      =  "<" *(%x20-3D / %x3F-7E) ">"
                ; açılı ayraçları içermeksizin açılı
                ; ayraçlar içine alınmış BOŞKRK ve
                ; GÖRKRK dizgeleri

                ; son çare olarak düzyazı açıklama

```

5. Güvenlikle İlgili Hususlar

Güvenliğin bu belgeyle gerçekten de bir ilgisi yoktur.

A. NÜVE

Bu bölüm ABNF'nin kendine özgü dilbilgisinin çekirdeğini oluşturan kural tanımlarını içerir. Bu tanımlar çekirdek kural kümesi olarak kullanılabilir. (Buradaki kurallar RFC çevirilerinde kullanılmak üzere Türkçe'ye uyarlanmıştır; İngilizce RFC'lerle ilgili kural tanımları için bu belgenin İngilizce aslına bakınız.)

Çekirdek Kurallar

Belli başlı temel kurallar büyük harflidirler; BOŞ, YSEK, SSSB, RAKAM, HARF gibi.

HARF	=	%x41-5A / %x61-7A	; A-Z / a-z
BIT	=	"0" / "1"	
KRK	=	%x01-7F	; Hiçlik (\0) karakteri hariç ; 7 bitlik bir US-ASCII karakteri
CR	=	%x0D	; satırbaşı karakteri
CRLF	=	CR LF	; Genel Ağ standardı olarak satırsonu
DNT	=	%x00-1F / %x7F	; Denetim karakterleri
RAKAM	=	%x30-39	; 0-9
ÇTIRNAK	=	%x22	; " (Çift tırnak)
RAKAM16	=	RAKAM / "A" / "B" / "C" / "D" / "E" / "F"	
YSEK	=	%x09	; yatay sekme karakteri
LF	=	%x0A	; satırsonu karakteri
SÜRBOŞ	=	*(BOŞLUK / CRLF BOŞLUK)	; sürekli (satırsonunu aşan) boşluk
SEKİZLİ	=	%x00-FF	; 8 bitlik veri
BOŞKRK	=	%x20	; boşluk karakteri

```
GÖRKRK      =  %x21-7E
                ; görünür (basılabilir) karakterler

BOŞLUK      =  BOŞKRK / YSEK
                ; boşluklar
```

Ortak Kodlama

Harici olarak, veri, yüksek sekizinci biti sıfırlanmış sekiz bitlik alanda yedi bitlik US–ASCII olarak tanımlanan ve "ağa göre ASCII" denilen biçimde sunulur. Değer dizgesi, ağa önce gönderilen yüksek değerli baytların solda olduğu "ağ bayt sıralaması"na göre dizilir.

B. Teşekkür

ABNF sözdizimi ilk olarak RFC 733'te belirtilmişti. BNF gösteriminin daha küçük ve daha anlaşılır olan arttırımlı BNF'ye dönüştürülmesinden SRI International'dan Ken L. Harrenstien sorumludur.

Bu son proje, yazdıkları eposta belirtimi olmadığı halde belirtim yazarları tarafından tekrar tekrar atıf yapılan RFC 822'nin işe yarar parçalarını toplamak şeklinde basit bir çaba olarak, arttırımlı BNF'nin açıklanması adıyla başladı. Çalışma grubu, basitçe körlemesine mevcut metni ayrı bir belgeye aktarmak yerine mevcut belirtilimlerdeki ve son 15 yılın belirtilimlerdeki yararlılıkları kadar yetersizliklerini de dikkatle inceleme yolunu seçti ve belirtimi genişletmeye çalıştı. Bu, projeyi ilk düşünülenden daha gözü yükseklerde bir şeye dönüştürdü. İlginç olan, liste gösteriminin kaldırılması gibi sürpriz kararlardan dolayı sonucun özgün çalışmadan devasa bir fark içermemesidir.

Belirtimin şu an ki aşamasında önemli destekleriyle Robert Elz, Roger Fajman, Aviva Garrett, Tom Harsch, Dan Kohn, Bill McQuillan, Keith Moore, Chris Newman, Pete Resnick, Henning Schulzrinne ile DRUMS çalışma grubunun bir kısmı rol aldı.

C. Kaynakça

1. [US–ASCII]
Kodlu Karakter Kümesi — Bilgi Değişimi için 7 Bitlik Amerikan Standart Kodu — **Coded Character Set–7–Bit American Standard Code for Information Interchange** — ANSI X3.4–1986 — 1986
2. [RFC733]
ARPA Ağ Metin İletisi için Standart — **Standard for the Format of ARPA Network Text Message** — Crocker, D., Vittal, J., Pogran, K. ve D. Henderson — RFC 733 — Kasım 1977
3. [RFC822]
ARPA Genel Ağ Metin İletilerinin Biçimi için Standart — **Standard for the Format of ARPA Internet Text Messages** — Crocker, D. — STD 11, RFC 822 — Ağustos 1982

D. Yazar Adresleri

David H. Crocker

Paul Overell

Internet Mail Consortium
675 Spruce Dr.
Sunnyvale, CA 94086 USA

Demon Internet Ltd
Dorking Business Park
Dorking

Surrey, RH4 1HN
UK

Telefon: +1 408 246 8253
Faks: +1 408 249 6205
Eposta: <dcrocker (at) imc.org> <paulo (at) turnpike.com>

E. Tam Telif Hakkı Beyanı

Copyright © The Internet Society (1997). All Rights Reserved.

This document and translations of it may be copied and furnished to others, and derivative works that comment on or otherwise explain it or assist in its implementation may be prepared, copied, published and distributed, in whole or in part, without restriction of any kind, provided that the above copyright notice and this paragraph are included on all such copies and derivative works. However, this document itself may not be modified in any way, such as by removing the copyright notice or references to the Internet Society or other Internet organizations, except as needed for the purpose of developing Internet standards in which case the procedures for copyrights defined in the Internet Standards process must be followed, or as required to translate it into languages other than English.

The limited permissions granted above are perpetual and will not be revoked by the Internet Society or its successors or assigns.

This document and the information contained herein is provided on an "AS IS" basis and THE INTERNET SOCIETY AND THE INTERNET ENGINEERING TASK FORCE DISCLAIMS ALL WARRANTIES, EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO ANY WARRANTY THAT THE USE OF THE INFORMATION HEREIN WILL NOT INFRINGE ANY RIGHTS OR ANY IMPLIED WARRANTIES OF MERCHANTABILITY OR FITNESS FOR A PARTICULAR PURPOSE.

Notlar

Belge içinde dipnotlar ve dış bağlantılar varsa, bunlarla ilgili bilgiler bulundukları sayfanın sonunda dipnot olarak verilmeyip, hepsi toplu olarak burada listelenmiş olacaktır.

(B1) <ftp://ftp.rfc-editor.org/in-notes/bcp/bcp78.txt>

(B3) <http://www.ietf.org/>

Bu dosya (rfc2234.pdf), belgenin XML biçiminin T_EXLive ve belgeler-xsl paketlerindeki araçlar kullanılarak PDF biçimine dönüştürülmesiyle elde edilmiştir.

17 Ocak 2007